Using Residual Vectors in MSC/NASTRAN Dynamic Analysis to Improve Accuracy

presented at the 1991 MSC World Users' Conference by

> Ted Rose Staff Scientist MacNeal-Schwendler Corp

#### ABSTRACT:

When using a modal approach to perform a dynamic analysis, a number of approximations must be made. The most limiting of these is the assumption that the calculated mode shapes are capable of representing the full dynamic response of the structure. In the past, a number of methods have been used to improve on this approximation (see reference 2). One of these is Mode Acceleration which is available in MSC/NASTRAN. The drawback to Mode Acceleration is that an "equivalent static" solution must be done at each time step to account for the "static" response of the high frequency modes.

A more efficient method of accounting for high frequency modes is provided in this application note. This method consists of augmenting the modes by appending static solution vectors. The included DMAP orthogonalizes these vectors to the modes (if requested) and results in uncoupled equations of motion.

Unlike Mode Acceleration, which is applied only at the residual structure, this method may be applied to superelements to improve the quality of the dynamic reduction and analysis results.

#### Theory

The theory is presented for superelements using component modal synthesis, but applies for the residual structure solution(and non-superelement analysis).

In component modal synthesis<sup>1</sup> (CMS), there are a number of methods to reduce superelement matrices to the degrees of freedom. In MSC/NASTRAN, there are currently several options, static reduction, generalized dynamic reduction(GDR), or component modal synthesis(CMS)<sup>2</sup>. The theory shown below applies to all reduction methods.

# I "Fixed-boundary" Component Modal synthesis (default in MSC/NASTRAN)

# A. Description in MSC/NASTRAN matrix notation

The default method of performing component modal synthesis in MSC/NASTRAN is "fixed-boundary", often referred to as the Craig-Bampton<sup>1</sup> method. In order to perform the default component modal synthesis, the model is partitioned into two sets of degrees of freedom (dof). The first set, called the "T" set, represents the physical boundary points. The second set is the interior dof, called the "O" set.

First, a set of "constraint" modes is generated. Each "constraint" mode represents the motion of the superelement resulting from moving one boundary dof 1.0 unit while holding all other boundary dof constrained. Therefore, there will be one "constraint" mode for each boundary dof.

This problem is solved in matrix form as follows:

$$\begin{bmatrix} K_{\text{oo}} & | & K_{\text{ot}} \\ \hline --- & | & --- \\ K_{\text{to}} & | & K_{\text{tt}} \end{bmatrix} \begin{bmatrix} G_{\text{ot}} \\ \hline --- \\ I_{\text{tt}} \end{bmatrix} = \begin{bmatrix} 0 \\ P_{\text{t}} \end{bmatrix}$$
 (1)

where  $I_{tt}$  is an identity matrix representing unit displacements for the T-set and  $G_{ot}$  represents the motion of the interior dof resulting from the boundary motion(note:  $P_t$  is not actually applied).

The first line from equation (1) gives:

$$\{G_{ot}\} = -[K_{oo}]^{-1}[K_{ot}]\{I_{tt}\}$$
 (2)

which results in the following "constraint" modes:

$$\{G_{t}\} = \begin{bmatrix} G_{ot} \\ --- \\ I_{tt} \end{bmatrix}$$
 (3)

which are identical to the vectors in Guyan reduction.

Now, the "T" dof are constrained (partitioned out) and the "O" set equations are solved for the "fixed-boundary" modes of the component,  $\{G_{og}\}$ :

$$-w^{2}[M_{oo}]\{G_{oq}\} + [K_{oo}]\{G_{oq}\} = 0$$
 (4)

where  $\{G_{\text{Oq}}\}$  represents the "fixed-boundary" modes. The number of "fixed-boundary" modes is determined by user input on EIGR or EIGRL entries. These modes are concatenated onto the "constraint" modes to form the transformation to the generalized coordinates:

$$\{U_{gen}\} = \begin{bmatrix} G_{ot} & G_{oq} \\ --- & --- \\ I_{tt} & 0 \end{bmatrix}$$
 (5)

The mass and stiffness matrices are now pre- and post-multiplied by these vectors to obtain the "generalized" mass and stiffness for the component:

$$[K_{aa}] = \{U_{gen}\}^{T}[K_{ff}]\{U_{Gen}\}$$
 (6)

$$[M_{aa}] = {U_{Gen}}^{T}[M_{ff}]{U_{Gen}}$$
 (7)

(where the "F" set is the union of the "T" and "O" sets)

These "generalized" matrices contain physical dof (the T-set) representing the boundaries and "modal" dof (the Q-set) representing the "fixed-boundary" component modes.

At this point, these matrices can be treated like any other structural matrices, and data recovery can be performed for the component in a manner similar to that used for solutions in modal coordinates. That is, the displacements of the generalized coordinates are multiplied by the associated vectors and added together to get the physical displacements.

An approximation is made that the limited number of component modes obtained can adequately represent the dynamic response of the component. There is no set rule for the number of component modes to obtain, and unless all modes of the component are obtained, the high-frequency response is ignored.

# B. Residual Vectors in component modal synthesis:

In order to improve the accuracy of the dynamic solution, additional shape functions for the superelement will be concatenated onto the CMS shape functions. These new shape functions consist of static solutions for the superelement resulting from applied loads which resemble the dynamic loads.

During superelement reduction (normally Sol 103), the LOADSET-LSEQ option can be used to assemble static loading vectors for the superelements (this is not normally done in Sol 103, but is done when the DMAP is used). These loading vectors can be reduced to 0-set (the interior dof) size and the 0-set static problem can be solved for the static (fixed-boundary) response of the superelement. Since  $K_{\text{OO}}$  has already been inverted during the static reduction phase, this step is efficient. All that needs to be done is the following:

$$\{U_{\text{ores}}\} = [K_{\text{oo}}]^{-1} \{P_{\text{o}}\}$$
 (8)

where U ores is the set of shape functions representing the fixed-boundary static response of the superelement to the applied loads. If the residual vectors are not linear combinations of the "constraint" and "fixed-boundary" modes shown in equation (5), they may be concatenated to the modes to form the new generalized coordinates:

$$\{U_{\text{gennew}}\} = \begin{bmatrix} G_{\text{ot}} & G_{\text{oq}} & U_{\text{ores}} \\ --- & --- & --- \\ I_{\text{tt}} & 0 & 0 \end{bmatrix}$$
 (9)

These new generalized coordinates are then treated in the same manner as before.

An option to orthogonalize the resultant vectors,  $\mathbf{U}_{\text{ores}}$ , to the component modes,  $\mathbf{G}_{\text{oq}}$ , is included. In this case, an intermediate set of generalized

coordinates are used to reduce the matrices and a new eigenvalue problem is solved using the reduced mass and stiffness:

$$\{U_{\text{Geno}}\} = \{G_{\text{og}} \mid U_{\text{ores}}\}$$
 (10)

$$[K_{\text{Geno}}] = \{U_{\text{Geno}}\}^{T}[K_{\text{OO}}]\{U_{\text{Geno}}\}$$
(11)

$$[M_{Geno}] = {U_{Geno}}^{T}[M_{oo}]{U_{Geno}}$$
(12)

$$-w_{\text{onew}}^{2}[M_{\text{Geno}}]\{U_{\text{onew}}\} + [K_{\text{Geno}}]\{U_{\text{onew}}\} = 0 \quad (13)$$

The solution of this eigenvalue problem will result in the original component modes, plus one new (high frequency) pseudo-mode corresponding to each column in the loading matrix. These new mode shapes  $(U_{\text{onew}})$ , when multiplied by the intermediate generalized vectors  $(U_{\text{Geno}})$ , result in a new set of generalized vectors  $(G_{\text{oqnew}})$  which are orthogonal.

These new vectors are substituted for the original component modes and residual vectors:

$$\{G_{oa}\} = \begin{bmatrix} U_{ot} & G_{oqnew} \\ --- & --- \\ I_{tt} & 0 \end{bmatrix}$$
 (14)

and the mass, stiffness, and damping are reduced as before.

If this option is used, the resultant "residual" vectors will be orthogonal to the component modes.

# II. USING RESIDUAL VECTORS IN A MODAL ANALYSIS

When performing a modal solution for the residual structure, the A-set matrices for are reduced to the H-set of modal coordinates in a manner similar to CMS. The difference is that there are no static reduction vectors (constraint modes). As in component modes, residual vectors can be concatenated onto the system modes and orthogonalized to them. This results in a set of modal equations which include the low frequency modes found during the original eigensolution and a representation of the high frequency content of the static loading matrix (residual vectors).

## IMPLEMENTATION:

Both of the above methods are implemented in a DMAP alter for version 66 of MSC/NASTRAN running on a Vax (Modifications may be necessary for different versions and computers). The alter is controlled by a PARAMETER named RESFLEX.

# PARAM, RESFLEX, x

If x = 0 then don't perform residual flexibility.

If x = -1 then perform residual flexibility on the superelements, but don't orthogonalize the vectors. Perform residual flexibility at the residual structure and perform orthogonalization.

If x = -2 then same as x = -1, except perform orthogonalization at the superelement level.

The DMAP requires that loading vectors (using LOADSET-LSEQ) must be included in the original run for CMS and system modes. If orthogonalization is requested, then the DMAP also requires a second SUBCASE for each superelement and the residual structure with a METHOD command pointing to an EIGR or EIGRL entry requesting all eigenvalues be found (shown in samples).

### SAMPLE PROBLEMS:

#### Problem 1: Modal transient

A simple beam problem is used (figure 1). A transient analysis is performed and the results are compared using (a) modal transient with all modes, (b) modal transient with modes up to 3 hz, (c) modal transient with modes to 3 hz and Mode Acceleration, and (d) modal transient with modes to 3 hz using residual vectors. Results are shown in figures 2 and 3. The data decks are listed in the appendix.

# Problem 2: Superelement component modes

A model of a "two-headed flyswatter" (figure 4) is used. It is divided into three superelements and a residual structure. System modes under 200 hz are compared using residual structure only (benchmark), (a)static reduction for the superelements, (b)CMS to 100hz for the superelements, and (c)CMS to 100hz and residual vectors for the superelements. The results are presented in the following table.

TABLE 1

COMPARISON OF FREQUENCIES USING DIFFERENT REDUCTION METHODS
SYSTEM MODES FOUND UNDER 200 HZ

MODE	SOLVED AS R.S. ONLY	(A) S.E. STATIC CONDENSATION	(B) S.E. USING CMS TO 100HZ	(C) S.E CMS + RES VECTORS
1 2 3 4 5 6 7 8	5.482 9.493 32.555 39.885 55.507 78.164 78.615	5.898 12.165 84.460 84.954 88.682 147.575	5.482 9.493 32.646 39.888 60.711 84.954 100.757	5.482 9.493 32.568 39.886 56.086 78.800 79.862 185.907
9	191.581	l		195.328

#### CONCLUSIONS:

Residual vectors can be used in any structured dynamic solution where CMS and/or modal solutions are used. Since the method provides improved dynamic representation of the superelement or system, it will result in improved answers with a minimum of user effort and cost. This approach is not only more efficient than mode acceleration but also works for superelements.

A drawback to the method is that the loadings must be known in advance in order to include their effect in the CMS or

modes. However, since the residual vector DMAP does not take advantage of automatic restart, any approximate loading vectors may be used in the reduction process, resulting in improved dynamic representation of the problem for a minimal increase in cost.

The DMAP does not utilize automatic restart in that a change in loading vectors will not automatically cause dynamic reduction to be performed again. However, if the DMAP is present and a change in the model or Case Control causes dynamic reduction to be performed, then new residual vectors based on the current loads will be calculated during the reduction process .

Also, since the residual vectors are treated as modes, they will have associated modal mass and damping. No special effort is required to handle the modal mass, but the modal damping may require special consideration. The pseudo modes will be high frequency and will have the effects of any physical damping in the model plus any modal damping specified using a TABDMP1. It is recommended as a minimum to add a small amount of modal damping for these pseudo modes. If no damping is included for these pseudo modes, then the results may exhibit some high frequency effects at the frequencies of the pseudo modes.


- Coupling of Substructures for Dynamic Analysis, R. R. Craig and M. C. C. Bampton, AIAA Journal, vol 6, no 7, July, 1968.
- MSC/NASTRAN User's Manual, Mike Reymond, Editor, The MacNeal-Schwendler Corporation, MSR-39, November, 1988.

```
$
 $
     RESIDUAL VECTOR ALTER - FILE NAME IS RESFLEX.ALT
 $ USE LOADSET/LSEQ COMBINATIONS TO GENERATE STATIC SOLUTIONS FOR THE
       O-SET FOR EACH S.E., THEN CONCATENATE THEM ONTO THE COMPONENT
        MODES
    FOR EACH SUPERELEMENT AND THE RESUDUAL IF RESIDUAL VECTORS
        ARE REQUESTED, THEN A SECOND SUBCASE MUST APPEAR WITH A
       METHOD COMMAND REFERENCING AN EIGR ENTRY REQUESTING THAT
       ALL MODES BE FOUND. THIS SECOND EIGENVALUE SOLUTION (NOT A FULL
       EIGENVALUE SOLUTION - IT IS WORKING WITH REDUCED EQUATIONS) WILL
       ORTHOGONALIZE THE RESIDUAL VECTORS TO THE COMPONENT MODES.
 $ FOR STRUCTURED SOL
 COMPILE PHASE1DR, SOUIN=MSCSOU, NOLIST, NOREF
 ALTER 2 $
 TYPE PARM, , I , Y , RESFLEX=0 $
ALTER 42
 IF (RESFLEX<0) SELG=ITOL(RESFLEX) $
 COMPILE SEMRM, SOUIN-MSCSOU, LIST, NOREF
 ALTER 39,39
 CALL SEMR3 GOAT, CASES, KFF, MFF, DYNAMICS, USET, SILS, EQEXINS,
      GPLS, BGPDTS, CSTMS, LOO, MOA1/
      PHIAZ, PHIOZ, CMLAMA, KLAA, GOAQ, MLAA1/
      ERROR/EPSMALC/EPSMALU/MAXRATIO/EPSBIG/INRLM/EPSRC/
      ASING/PRPHIVZ/NOASET/NORC/NOQSET/NOTSET/NORSET/
      NOOSET/SEID/METHOD/DYNRED/NOUP $
COMPILE SEMR3, SOUIN=MSCSOU, LIST, NOREF
ALTER 1,1
SUBDMAP SEMR3
                GOAT, CASES, KFF, MFF, DYNAMICS, USET, SILS, EQEXINS,
                GPLS, BGPDTS, CSTMS, LOO, MOA1/
                PHIAZ, PHIOZ, CMLAMA, KLAA, GOAQ, MLAA1/
                ERROR/EPSMALC/EPSMALU/MAXRATIO/EPSBIG/INRLM/EPSRC/ASING/
                PRPHIVZ/NOASET/NORC/NOQSET/NOTSET/NORSET/NOOSET/SEID/
                METHOD/DYNRED/NOUP $
TYPE PARM,,I,Y,RESFLEX=0 $
TYPE PARM,,I,N,NEED,EXTRACOL,NOUP $
TYPE DB, USETB, GM $
TYPE DB, PJ, SLIST, EMAP, PA, MAPS $
ALTER 105 $ AFTER INREL
IF (RESFLEX<>0) THEN $ E RES VECTOR ALTER
  MESSAGE //'RESIDUAL VECTORS REQUESTED FOR SE'/SEID $
$ RESIDUAL VECTORS
$
$ CHECK IF GOQ HAS SPACE FOR RES VECTORS
 MATMOD GOQ,,,,,/GOQNULL,/12/S,N,NULCGOQ $
MESSAGE //'NULL COL IN GOQ ='/NULCGOQ $
$ ASSEMBLE PG FOR USE IN RESIDUAL VECTORS
  IF (NOUP >= 0 ) THEN $ (A6)
    DBVIEW PAUP1 = PA
                            (WHERE SEID=* AND WILDCARD=TRUE) $
    DBVIEW MAPUPL1 = MAPS (WHERE SEID=* AND WILDCARD=TRUE) $
    SELA PJ, SLIST, EMAP, EQEXINS, PAUP1, MAPUPL1/PG/SEID/'SEID'/
               0/S,N,NOPGS $
  ELSE $ (A6)
    EQUIVX PJ/PG/ALWAYS $
 ENDIF $ (A6)
 PARAML PG//'TRAILER'/1/S,N,NPG $ # OF COL IN PG
 MATMOD PG,,,,,/PGNULL,/12/S,N,NULCPG $
IF (NULCPG >= 1 ) THEN $ D REMOVE NULL COLUMNS FROM PG
```


```
PARTN PG,PGNULL,/PGRES,,,/1 $
      NPG = NPG-NULCPG $
   ELSE $ D
     EQUIVX PG/PGRES/ALWAYS $
   ENDIF $ D END OF REMOVE NULL COLUMNS FROM PG
    IF (NPG<1) THEN $ C
      MESSAGE //'WARNING - RESIDUAL VECTORS REQUESTED'/
         ' BUT NO LOADING VECTORS AVAILABLE FOR SE'/SEID $
   ELSE $ C HAVE LOADING VECTOR
      NOOP() $ ADDED FOR V66 ERROR
      IF (NPG>NULCGOQ) THEN $ B
    NO SPACE IN GOO
        NEED=NPG-NULCGOQ $
        MESSAGE //'RESIDUAL VECTORS REQUESTED, BUT INSUFFICIENT'/
             ' Q-SET DOF AVAILABLE FOR SE'/SEID/' NEED'/NEED/' MORE'/
               , RES VECTORS NOT DONE' $
     ELSE $ B CALC RES VECTORS AND APPEND TO GOQ
 $
       ENOUGH DOF IN GOQ
       PARAML 'USET//'USET'///S,N,NOUSETX//'A'/S,N,NOASET/
                                               'M'/S,N,NOMSET/
                                              'O'/S,N,NOOSET/
                                              'S'/S,N,NOSSET $
       EQUIVX PGRES/PGN/NOMSET $
        IF (NOMSET>0) MCE2 USETB,GM,PGRES,,,/PGN,,, $
       EQUIVX PGN/PGO/NOSSET $
        IF (NOSSET>=0) UPARTN USET,PGN/PGO,,,/'N'/'O'/'A'/1 $
 $
        USE STATIC LOAD VECTORS TO SOLVE FOR RESIDUAL VECTORS
 $
       SSG3 LOO,,KOO,PGO,,,,/UORSFLX,,,/1/0/1/S,N,EPSI $
NORM UORSFLX/UORSFLX1/NPG/NOOSET//1 $
       EXTRACOL = NULCGOQ-NPG $ AVAILABLE COLUMNS
       IF (EXTRACOL>0) THEN $ A
 $
        APPEND NULL COLUMNS TO UORSFLX1 TO GET SIZE RIGHT
         MATGEN USET/XTRACOLA/11/1/EXTRACOL $
         UMERGE USET,XTRACOLA,/XTRACOLG/'G'/'A'/'O' $
         UPARTN USET,XTRACOLG/XTRACOLO,,,/'G'/'O'/'A'/1 $
         APPEND UORSFLX1,XTRACOLO/UORESFLX/1 $
       ELSE $ A
$
         NO EXTRA COL
$
         EQUIVX UORSFLX1/UORESFLX/ALWAYS $
       ENDIF $ A
$ MERGE RES VECTORS INTO GOQ
$
       PARTN GOQ, GOQNULL, /GOQ1,,,/1 $
       MERGE GOQ1, ,UORESFLX, ,GOQNULL, /GOQR/1 $
       IF (RESFLEX=-2) THEN $ ORTHOGONALIZE VECTORS
$ ADD SECOND EIGENSOLUTION TO GET INDEPENDENT VECTORS
        MATMOD GOQR,,,,,/GOQRNULL,/12/S,N,NULCGOQR $
PARTN GOQR,GOQRNULL,/GOQR1,,,/1 $
         SMPYAD GOGR1, MOO, GOGR1, , , /MOONEW/3///-1 $
        SMPYAD GOQR1,KOO,GOQR1,,,/KOONEW/3////-1 $
PARAML KOONEW//'TRAILER'/1/S,N,NOVEC $ # OF COL IN KOONEW
$ SOLVE NEW EIGENVALUE PROBLEM
        MESSAGE //'SOLVING SECOND EIGENVALUE PROBLEM' $
        CASE CASES,/CASERS/'TRAN'/2 $ GET SECOND RECORD
        IF (LANCZOS>-1) THEN $
          READ KOONEW, MOONEW, , , DYNAMICS, , CASERS/NEWLAMA, NEWPHIX,
                NEWMMI, NEIGVMAT/'MODE'/S,N, NOVEC $
        ELSE $
```

# ILLUSTRATIVE EXAMPLE FOR TRANSIENT RESPONSE ANALYSIS USING MSC/NASTRAN

• Elastic bar



Applied loading is —









Figure 2 Results of Beam Model Analyses



Close-up of Results from Beam Model Analyses




Figure 4 Two-Headed Flyswatter Model

```
REIGL KOONEW, MOONEW, DYNAMICS, CASERS, , , , / NEWLAMA, NEWPHIX,
                 NEWMMI, NEIGVMAT, / MODE //S, N, NOVEC $
$ PRINT RESULTS OF SECOND EIGENSOLUTION
$
          OFP NEIGVMAT, NEWLAMA, NEWPHIX// $
$ GET NEW MODAL COORDINATES - INCLUDING RES VEC - ORTHOGONALIZED
          MPYAD GOQR1, NEWPHIX, / GOQORTH $
          PARAML GOQ//'TRAILER'/1/S,N,NCGOQ $ # OF COL IN GOQ
         EXTRACOL = NCGOQ-NOVEC $ UNUSED COLUMNS
$
$
        APPEND NULL COLUMNS TO GOGORTH TO GET SIZE RIGHT
$
            MATGEN USET/XTRACLA/11/1/EXTRACOL $
           UMERGE USET,XTRACLA,/XTRACLG/'G'/'A'/'O' $
            UPARTN USET, XTRACLG/XTRACLO, , , /'G'/'0'/'A'/1 $
            APPEND GOQORTH, XTRACLO/GOQORTH1/1 $
            EQUIVX GOOORTH1/GOO/ALWAYS $
       ELSE $
         EQUIVX GOGORTH/GOG/ALWAYS $
       ENDIF $
     ENDIF $ B END OF CALC RES VECTORS AND APPEND TO GOQ
  ENDIF $ C END OF HAVE LOADING VECTOR
ENDIF $ E END OF RESIDUAL VECTOR ALTER
COMPILE MODERS, SOUIN-MSCSOU, LIST, NOREF $
ALTER 5
TYPE PARM,,I,Y,RESFLEX=0 $
TYPE DB,PA $
TYPE DB, USETB $
ALTER 86
$ ADD RESIDUAL VECTORS FOR RESIDUAL STRUCTURE
IF (RESFLEX<0) THEN $ (A0) RESIDUAL FLEXIBILITY
  MESSAGE //'RESIDUAL VECTORS REQUESTED' $
  PARAML PA//'TRAILER'/1/S,N,NPA $ NO OF COL IN PA
  MESSAGE //'NO OF COL IN PA ='/NPA $
  IF (NPA>0) THEN $ (A1) HAVE PA
  MATMOD PA,,,,,/PANULL,/12/S,N,NULCPA $ CHECK FOR NULL COL IF (NULCPA>O) THEN $ (A2) GET RID OF NULL COLUMNS
       PARTN PA, PANULL, /PARES, , , /1 $
       NPA=NPA-NULCPA $
    ELSE $ (A2)
       EQUIVX PA/PARES/ALWAYS $
     ENDIF $ (A2)
     IF (NPA<=0) THEN $ (A5)
        MESSAGE //'LOAD VECTOR FOR RESIDUAL STRUCTURE'/
               ' IS EMPTY, RESIDUAL VECTOR CALCULATION NOT'/
               ' PERFORMED' $
    ELSE $ (A5)
       DCMP USET, SILS, EQEXINS, MKAA/LAA, UAA/-1/0/BAILOUT/
          MAXRATIO/'A'/1.E-20/////s,N,SING/S,N,NBRCHG/
          S,N,ERR $
       SSG3 LAA,,MKAA,PA,,,,/UARSFLX,,,/1/0/1/S,N,EPSI $
NORM UARSFLX/UARSFLX1/NPA/NOASET//1 $
       APPEND PHIA, UARSFLX1/UARESFLX/1 $
$ FOR MODAL TRANSIENT, ALWAYS NORMALIZE RES FLEX VECTORS
      SMPYAD UARESFLX,MMAA,UARESFLX,,,/MAANEW/3///-1 $
SMPYAD UARESFLX,MKAA,UARESFLX,,,/KAANEW/3///-1 $
MESSAGE //SOLVING SECOND EIGENVALUE PROBLEM TO ORTHOGONALIZE'/
                ' RESIDUAL VECTORS' $
       CASE CASES,/CASERS/'TRAN'/2 $ GET SECOND SUBCASE FOR EIGENSOLN
      PURGEX /LAMA, PHIA, , , /ALWAYS $
IF (LANCZOS > -1) THEN $ (A4)
         READ KAANEW, MAANEW, , , DYNAMICS, , CASERS/LAMA, NEWPHIX,
              NEWMMI, NEIGVMAT/'MODE'/S, N, NOVEC $
      ELSE $ (A4)
```

```
REIGL KAANEW, MAANEW, DYNAMICS, CASERS, , , , / LAMA, NEWPHIX, NEWMMI, NEIGVMAT, / 'MODE'/S, N, NOVEC $
      ENDIF $ (A4)
$
$ PRINT RESULTS OF SECOND EIGENSOLUTION
$
      OFP NEIGVMAT, LAMA, NEWPHIX// $
      MPYAD UARESFLX, NEWPHIX/PHIA $
$
      MPYAD PHIA, PA, /MODEFORC/1 $ *******
       MATPRN MODEFORC// $ ******
    ENDIF $ (A5)
  ELSE $ (A1)
    MESSAGE //'NO LOAD VECTOR FOR RESIDUAL STRUCTURE'/
             ' RESIDUAL FLEXIBILITY CALCULATION NOT PERFORMED' $
  ENDIF $ (A1)
ENDIF $ (A0)
COMPILE SELR, SOUIN-MSCSOU, LIST, NOREF
ALTER 2
TYPE PARM,, I, Y, RESFLEX=0 $
ALTER 35,35
IF(SEID<>0 OR APP='STATICS ' OR APP='BKL0
                                              ' OR RESFLEX<0)THEN $
$ LOAD REDUCTION
$ALTER 86 $
SMATPRN PA// $
$ END OF ALTER FOR RESIDUAL VECTORS
$
```

### APPENDIX - FILES USED IN SAMPLE 1

```
$ FILE BULKDATA.DAT - BULK DATA COMMON TO ALL RUNS
$ PARAM POST SET FOR MSC/XL POST PROCESSING
PARAM, POST, 0
$ DEFINE GRID COORDINATES
GRID,1,,0.,0.,0.
=,*(1),=,*(12.),==
=(9)
$
$ DEFINE BEAM
CBAR, 101, 100, 1, 2, 0., 1., 0.
=,*(1),=,*(1),*(1),==
=(8)
PBAR, 100, 1, 2.185, 10.433, 3.535, 13.968
MAT1,1,1.0E+5,..3,1.E-4

COMM2,201,1,,,,,+CM01

+CM01,9.11E-3

COMM2,202,2,,,,+CM02

=,*(1),*(1),,,,*(1)
=(7)
+CM02,18.22E-3
*(1),=
=(7)
CONM2,211,11,,,,,,+CM11
+CM11,9.11E-3
$ EIGENVALUE EXTRACTION METHOD
EIGR,99,MGIV,0.,10000.,,,,+EIG
+EIG, MASS
$ CONSTRAIN BASE
SPC, 100, 1, 123456
$ GRADUAL STEP FORCE HISTORY
DLOAD,100,1.,1.,110,-1.,120
TLOAD2,110,100,,0,0,100.,0.,0
TLOAD2,120,100,,0,0,100.,0.,0-,+TL1
+TL1,-3.333,0.
LSEQ,999,100,10
FORCE, 10,4,,1.,,1.
TSTEP, 100,500,.01,1
ID DYNAMICS, TWO
SOL 112
TIME 5
CEND
TITLE= TUBE MODEL, MODAL SOLUTION, FIXED BASE
SUBTITLE= ALL MODES
SPC= 100
LOADSET = 999
DISP(PLOT)=ALL
DLOAD= 100
TSTEP= 100
METHOD= 99
BEGIN BULK
INCLUDE BULKDATA.DAT
ENDDATA
```

```
ID DYNAMICS, TWO
SOL 112
TIME 5
CEND
TITLE= TUBE MODEL, MODAL SOLUTION, FIXED BASE
SUBTITLE= MODES BELOW 10 HERTZ
SPC= 100
LOADSET = 999
DISP(PLOT)=ALL
DLOAD= 100
TSTEP= 100
METHOD= 999
BEGIN BULK
INCLUDE BULKDATA.DAT
EIGRL,999,0.,10.
ENDDATA
ID DYNAMICS, TWO
SOL 112
TIME 5
CEND
TITLE= TUBE MODEL, MODAL SOLUTION, FIXED BASE
SUBTITLE= MODES BELOW 10 HERTZ + MODAL ACCELERATION
SPC= 100
LOADSET = 999
DISP(PLOT) = ALL
DLOAD= 100
TSTEP= 100
METHOD= 999
BEGIN BULK
INCLUDE BULKDATA.DAT
PARAM, MODACC, 0
PARAM, DDRMM, -1
EIGRL,999,0.,10.
ENDDATA
ID DYNAMICS, TWO
SOL 112
TIME 5
$
$
    RESIDUAL VECTOR ALTER
INCLUDE RESFLEX.ALT
CEND
TITLE= TUBE MODEL, MODAL SOLUTION, FIXED BASE
SUBTITLE = MODES BELOW 10 HERTZ - RES FLEX ADDED
SPC= 100
LOADSET = 999
SUBCASE 1
DISP(PLOT) = ALL
DLOAD= 100
TSTEP= 100
METHOD= 9
SUBCASE 2
DLOAD= 100
TSTEP= 100
METHOD= 99
BEGIN BULK
INCLUDE BULKDATA.DAT
EIGR,9,MGIV,0.,10.
$ REQUEST RESIDUAL VECTORS
PARAM, RESFLEX, -2
ENDDATA
```